

EE671: VLSI DESIGN

SPRING 2024/25

LAXMEESHA SOMAPPA
DEPARTMENT OF ELECTRICAL ENGINEERING
IIT BOMBAY
laxmeesha@ee.iitb.ac.in



LECTURE – 20

DIGITAL SYNTHESIS


LOGIC SYNTHESIS: STATE MACHINE GUIDELINES

- Draw a state diagram.
- Label the states.
- Allocate state encoding.
- Label the transition conditions.
- Label the output values.
- Use parameters for the state variables.
- Use two procedures (one clocked for the state register and one combinational for the next state logic and the output decode logic).
- Use a case statement in the combinational procedure.
- Have a reset strategy (asynchronous or synchronous).
- Use default assignments and then corner cases.
- Keep state machine code separate from other code (i.e. don't mix other logic in with the state machine clocked and combinational procedures).



LOGIC SYNTHESIS: STATE MACHINE GUIDELINES

State Diagram




LOGIC SYNTHESIS: STATE MACHINE GUIDELINES

```
module state_machine (clock,reset,start,transmit,wait,stop,ack,offline,online);  
  // parameter declarations  
  parameter pIDLE    = 2'b10; // state labels and state encoding  
  parameter pRUN      = 2'b00;  
  parameter pPAUSE    = 2'b01;  
  parameter pFINISH   = 2'b11;  
  
  // IO declaration section  
  input clock;  
  input reset;  
  input start, transmit, wait, stop;  
  output ack, offline, online;  
  // internal variables declaration section  
  reg [1:0] state, next_state;  
  reg ack, offline, online;  
  
  // clocked procedure with synchronous reset  
  always @ (posedge clock)  
  if (reset) // reset strategy  
    state <= pIDLE;  
  else  
    state <= next_state;
```



LOGIC SYNTHESIS: STATE MACHINE GUIDELINES

```
// combinational procedure with case statement and output logic
always @ (start or transmit or stop or wait or state)
begin
    next_state = state; // default assignment to state and output variables
    ack      = 1'b0;
    offline  = 1'b0;
    online   = 1'b1;
    case (state)
        pIDLE:
            begin
                offline = 1'b1;
                online  = 1'b0;
                if (start)
                    next_state = pRUN;
            end
        pRUN:
            begin
                if (wait)
                    next_state = pPAUSE;
                if (stop)
                    next_state = pFINISH; // this has priority over the wait transition
            end
        pPAUSE:
            begin
                ack      = 1'b1;
                if (transmit)
                    next_state = pRUN;
                if (stop)
                    next_state = pFINISH;
            end
        pFINISH:
            begin
                ack = 1'b1;
            end
    endcase
end
endmodule
```



LOGIC SYNTHESIS: BLOCKING VS NON-BLOCKING

Refer to “Nonblocking Assignments in Verilog Synthesis; Coding Styles That Kill!” by Cliff Cummings – uploaded on Moodle

Nonblocking Assignments in Verilog Synthesis; Coding Styles That Kill!

by Cliff Cummings
Sunburst Design, Inc.

Abstract

One of the most misunderstood constructs in the Verilog language is the nonblocking assignment. Even very experienced Verilog designers do not fully understand how nonblocking assignments are scheduled in an IEEE compliant Verilog simulator and do not understand when and why nonblocking assignments should be used. This paper details how Verilog blocking and nonblocking assignments are scheduled, gives important coding guidelines to infer correct synthesizable logic and details coding styles to avoid Verilog simulation race conditions.



LOGIC SYNTHESIS

- ❑ Popular logic synthesis tools:
 - ❑ Synopsys: Design Compiler (DC)
 - ❑ Cadence: Genus
 - ❑ Open Source: Yosys (will be used in this course)

- ❑ One of the widely used scripting language in the industry:
 - ❑ Tcl: Tool Command Language
- ❑ Synopsys DC, Cadence Genus and Yosys all support tcl based scripting flow



LOGIC SYNTHESIS: YOSYS EXAMPLE TCL SCRIPT

```
yosys -import  
set design "and2"  
  
# read design  
read_verilog ../Verilog/$design.v  
hierarchy -check -top $design  
  
# high-level synthesis  
procs; opt; memory; opt; fsm; opt  
  
write_verilog -noattr ../$design\_rtl.v  
  
# show  
show -format ps -prefix ../$design\_rtl
```

Defining a variable called design and assigning it a string value "and2"

read_verilog <path_to_file>
-- to read-in the design file

Elaborate the design hierarchy with top cell name

Convert processes (internal behavioural structures) into multiplexers and registers

First level of basic optimizations & clean-up



LOGIC SYNTHESIS: YOSYS EXAMPLE TCL SCRIPT

```
yosys -import  
  
set design "and2"  
  
# read design  
read_verilog ../Verilog/$design.v  
hierarchy -check -top $design  
  
# high-level synthesis  
procs; opt; memory; opt; fsm; opt  
  
write_verilog -noattr ../$design\_rtl.v  
  
# show  
show -format ps -prefix ../$design\_rtl
```

Analyze memories and create circuits to implement them (using FFs or other custom IPs)

Another round of basic optimization & cleanup

Analyze and optimize finite state machines

Write the output RTL file. This is not yet technology mapped (Std. cell library independent)

Write a pictorial representation of the logic (can be .ps, .png formats) for user to view



LOGIC SYNTHESIS: YOSYS TCL SCRIPT WITH STD CELL LIB

```
yosys -import

set design "and2"

# read PDK verilog
read_verilog -Idir ../../../../sky130_fd_sc_hd/verilog
read_verilog -lib ../../../../sky130_fd_sc_hd/verilog/sky130_fd_sc_hd.v
read_verilog -DNO_PRIMITIVES -lib ../../../../sky130_fd_sc_hd/verilog/primitives.v

#Read Liberty
read_liberty -lib ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd_tt_025C_1v80.lib

# read design
read_verilog ../Verilog/$design.v
hierarchy -check -top $design

# high-level synthesis
procs; opt; memory; opt; fsm; opt

# low-level synthesis
techmap; opt

# map to target architecture
dfflibmap -liberty ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd_tt_025C_1v80.lib
abc -liberty ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd_tt_025C_1v80.lib

# cleanup
clean

# write synthesized design
write_verilog -noattr $design\_map.v

# show
show -format ps -lib ../../../../sky130_fd_sc_hd/verilog/sky130_fd_sc_hd.v -prefix ../$design\_map
```



LOGIC SYNTHESIS: YOSYS TCL SCRIPT WITH STD CELL LIB

```
yosys -import
```

```
set design "and2"
```

Pointing to the standard cell library

```
# read PDK verilog  
read_verilog -Idir ../../../../sky130_fd_sc_hd/verilog  
read_verilog -lib ../../../../sky130_fd_sc_hd/verilog/sky130_fd_sc_hd.v  
read_verilog -DNO_PRIMITIVES -lib ../../../../sky130_fd_sc_hd/verilog/primitives.v
```

```
#Read Liberty  
read_liberty -lib ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd__tt_025C_1v80.lib
```

```
# read design  
read_verilog ../Verilog/$design.v  
hierarchy -check -top $design
```

```
# high-level synthesis  
procs; opt; memory; opt; fsm; opt
```



LOGIC SYNTHESIS: YOSYS TCL SCRIPT WITH STD CELL LIB

```
# low-level synthesis  
techmap; opt
```

Map large RTL cells (adder, multiplier etc.) to gates

```
# map to target architecture  
dfplibmap -liberty ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd_tt_025C_1v80.lib  
abc -liberty ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd_tt_025C_1v80.lib
```

```
# cleanup  
clean
```

```
# write synthesized design  
write_verilog -noattr $design\_map.v
```

```
# show  
show -format ps -lib ../../../../sky130_fd_sc_hd/verilog/sky130_fd_sc_hd.v -prefix ./$design\_map
```

Map all registers in the RTL to the Std. Cell Library

Map all logics in the RTL to the Std. Cell Library



LOGIC SYNTHESIS: YOSYS MANUAL & DEMO

- ❑ If you want to read up more on Yosys:
https://yosyshq.net/yosys/files/yosys_manual.pdf
- ❑ You can install Yosys on your machines if you are interested
<https://yosyshq.net/yosys/download.html>
- ❑ Class Demo on few circuits – Yosys Examples
- ❑ You will see later that “Openlane” which is an open source RTL to GDSII flow wraps all open source tools including Yosys with Python wrappers to seamlessly integrate multiple tools under one banner
 - ❑ You don’t need to learn Tcl scripting – you will use Python to run Yosys later
 - ❑ The Tcl scripts provided here are very specific to Yosys and will not work on Synopsys DC and Genus.



LOGIC SYNTHESIS: EXAMPLE

- ❑ Sequence of steps:
 - ❑ Write Verilog Code
 - ❑ Write Testbench to test the Verilog code (\$dumpvars will dump vcd waveform)
 - ❑ Simulate code and tb in iverilog (Icarus Verilog)
 - ❑ Open waveform (.vcd) in gtkwave and check waveforms

- ❑ Synthesize Verilog code (Yosys) and check output RTL file (without map to std. cell library – just for your reference)
- ❑ Synthesize Verilog code (Yosys) and check output RTL file (with map to std. cell library – this is the main output RTL)
- ❑ Simulate the output RTL with earlier tb using Iverilog – to check it is functionally still the same



LOGIC SYNTHESIS: EXAMPLE DIRECTORY STRUCTURE

```
2.MUX2to1/
├── Synthesis
│   ├── synthesize_map_rtl.tcl
│   ├── synthesize_rtl.tcl
│   └── Verification
│       ├── mux2to1_TB
│       ├── mux2to1_TB.v
│       └── run-iverilog.sh
└── Verilog
    ├── mux2to1_TB.v
    ├── mux2to1.v
    └── run-iverilog.sh
```

- **2.Mux2to1 is the main folder**
- **In the Verilog folder, we have the main Verilog file, tb and run command for iverilog**
- **Inside Synthesis folder, we have tcl scripts for Yosys**
- **Inside Synthesis folder, we have Verification folder to re-run simulation with RTL netlist**



LOGIC SYNTHESIS: EXAMPLE MUX

```

mux2to1.v      mux2to1_TB.v
1 module mux2to1(Y, A, B, SEL);
2   output reg Y;
3   input  A, B, SEL;
4   //reg f;
5
6   always @(A or B or SEL)
7   begin
8     if (SEL)
9       Y = B;
10    else
11      Y = A;
12  end
13
14 endmodule
15

```

```

mux2to1.v      mux2to1_TB.v
1 timescale 1 ns / 1 ns
2
3 module main;
4   reg A, B, SEL;
5   reg[2:0] inputs;
6   wire Y;
7   integer idx;
8
9   mux2to1 dut(.Y(Y), .A(A), .B(B), .SEL(SEL));
10
11  initial
12  begin
13    for (idx = 0; idx <= 7; idx = idx + 1)
14    begin
15      inputs = idx;
16      SEL = inputs[2];
17      A = inputs[1];
18      B = inputs[0];
19      #10
20      $display("%t: SEL=%b, A=%b, B=%b, Y=%b", $time, SEL, A, B, Y);
21    end
22  end
23
24  initial
25  begin
26    $dumpfile("waveforms.vcd");
27    $dumpvars(0,dut);
28  end
29
30 endmodule
31

```




LOGIC SYNTHESIS: EXAMPLE MUX SIMS (IVERILOG)

```
run-iverilog.sh mux2to1_TB.v

1#!/bin/sh
2
3design="mux2to1"
4rm -rf ${design}_TB
5echo "Testing:" ${design}
6iverilog -o ${design}_TB ${design}.v ${design}_TB.v
7vvp ${design}_TB
```

Shell script to run iverilog: on the terminal execute:
./run-iverilog.sh



Open waveforms using GTKWave. On the terminal
gtkwave <path_vcd_file>



LOGIC SYNTHESIS: EXAMPLE (YOSYS)

```

synthesize_rtl.tcl  | synthesize_map_rtl.tcl
1 yosys -import
2
3 set design "mux2to1"
4
5 # read design
6 read_verilog ../Verilog/$design.v
7 hierarchy -check -top $design
8
9 # high-level synthesis
10 procs; opt; memory; opt; fsm; opt
11
12 write_verilog -noattr ./ $design\_rtl.v
13
14 # show
15 show -format ps -prefix ./ $design\_rtl
16

```




Yosys outputs (.v and .ps)

```

synthesize_rtl.tcl  | mux2to1_rtl.v
1 /* Generated by Yosys 0.45+139 (g
2
3 module mux2to1(Y, A, B, SEL);
4     input A;
5     wire A;
6     input B;
7     wire B;
8     input SEL;
9     wire SEL;
10    output Y;
11    wire Y;
12    assign Y = SEL ? B : A;
13 endmodule
14

```



mux2to1

Call Yosys on the terminal with the tcl script

```
yosys synthesize_rtl.tcl
```

LOGIC SYNTHESIS: EXAMPLE (YOSYS MAP TO LIBRARY)

```

synthesize_rtl.tcl      synthesize_map_rtl.tcl
1 yosys -import
2
3 set design "mux2to1"
4
5 # read PDK verilog
6 read_verilog -Idir ../../../../sky130_fd_sc_hd/verilog
7 read_verilog -lib ../../../../sky130_fd_sc_hd/verilog/sky130_fd_sc_hd.v
8 read_verilog -DNO_PRIMITIVES -lib ../../../../sky130_fd_sc_hd/verilog/primitives.v
9
10 #Read Liberty
11 read_liberty -lib ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd_tt_025C_1v80.lib
12
13 # read design
14 read_verilog ../Verilog/$design.v
15 hierarchy -check -top $design
16
17 # high-level synthesis
18 procs; opt; memory; opt; fsm; opt
19
20 # low-level synthesis
21 techmap; opt
22
23 # map to target architecture
24 dfflibmap -liberty ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd_tt_025C_1v80.lib
25 abc -liberty ../../../../sky130_fd_sc_hd/lib/sky130_fd_sc_hd_tt_025C_1v80.lib
26
27
28 # cleanup
29 clean
30
31 # write synthesized design
32 write_verilog -noattr $design\_map.v
33
34 # show
35 show -format ps -lib ../../../../sky130_fd_sc_hd/verilog/sky130_fd_sc_hd.v -prefix ../$design\_map
36


```



LOGIC SYNTHESIS: EXAMPLE (YOSYS MAP TO LIBRARY)

Call Yosys on the terminal with the tcl script

```
yosys synthesize_map_rtl.tcl
```



```
synthesize_map_rtl.tcl | mux2to1_map.v
1 /* Generated by Yosys 0.45+139 (g.
2
3 module mux2to1(Y, A, B, SEL);
4   input A;
5   wire A;
6   input B;
7   wire B;
8   input SEL;
9   wire SEL;
10  output Y;
11  wire Y;
12  sky130_fd_sc_hd__mux2_1 _0_ (
13    .A0(A),
14    .A1(B),
15    .S(SEL),
16    .X(Y)
17  );
18 endmodule
19
```



LOGIC SYNTHESIS: EXAMPLE ALU

```

alu.v      alu_TB.v
1 module alu(a, b, cin, sel, y);
2   input [7:0] a, b;
3   input cin;
4   input [3:0] sel;
5   output [7:0] y;
6   reg [7:0] y;
7   reg [7:0] arithval;
8   reg [7:0] logicval;
9   // Arithmetic unit
10  always @(a or b or cin or sel) begin
11      case (sel[2:0])
12          3'b000 : arithval = a;
13          3'b001 : arithval = a + 1;
14          3'b010 : arithval = a - 1;
15          3'b011 : arithval = b;
16          3'b100 : arithval = b + 1;
17          3'b101 : arithval = b - 1;
18          3'b110 : arithval = a + b;
19          default : arithval = a + b + cin;
20      endcase
21  end
22  // Logic unit
23  always @(a or b or sel) begin
24      case (sel[2:0])
25          3'b000 : logicval = ~a;
26          3'b001 : logicval = ~b;
27          3'b010 : logicval = a & b;
28          3'b011 : logicval = a | b;
29          3'b100 : logicval = ~(a & b);
30          3'b101 : logicval = ~(a | b);
31          3'b110 : logicval = a ^ b;
32          default : logicval = ~(a ^ b);
33      endcase
34  end
35  // Multiplexer
36  always @(arithval or logicval or sel) begin
37      case (sel[3])
38          1'b0 : y = arithval;
39          default : y = logicval;
40      endcase
41  end
42 endmodule

```

```

alu.v      alu_TB.v
1 [timescale 1 ns / 1 ns
2
3 module main;
4   reg [7:0] a, b;
5   reg cin;
6   reg [3:0] sel;
7   wire [7:0] y;
8   integer idx;
9
10  alu dut(.a(a), .b(b), .cin(cin), .sel(sel), .y(y));
11
12  initial
13  begin
14      // a, b are fixed through out the test
15      a = 8'h93;
16      b = 8'hA7;
17      for (idx = 0; idx <= 15; idx = idx + 1)
18      begin
19          // Generate sel
20          sel = idx;
21          // cin = 1 only for sel = 5
22          if (idx == 5)
23              cin = 1'b1;
24          else
25              cin = 1'b0;
26          // Apply a 10-time unit delay; think of it as the system machine cycle
27          #10
28          // Display results after 10 time units $
29          $display("%t: a=%h, b=%h, cin=%b, sel=%h, y=%h", $time, a, b, cin, sel, y);
30      end
31  end
32
33  initial
34  begin
35      $dumpfile("waveforms.vcd");
36      $dumpvars(0, dut);
37  end
38 endmodule
39

```



LOGIC SYNTHESIS: EXAMPLE ALU YOSYS OUTPUT

```

alu.v      alu_map.v
1  /* Generated by Yosys 0.45+139 (g
2
3  module alu(a, b, cin, sel, y);
4      wire _000_;
5      wire _001_;
6      wire _002_;
7      wire _003_;
8      wire _004_;
9      wire _005_;
10     wire _006_;
11     wire _007_;
12     wire _008_;
13     wire _009_;
14     wire _010_;
15     wire _011_;
16     wire _012_;
17     wire _013_;

```

Only snippets are
shown – full file is
uploaded on Moodle

```

227     input cin;
228     wire cin;
229     input [3:0] sel;
230     wire [3:0] sel;
231     output [7:0] y;
232     wire [7:0] y;
233     sky130_fd_sc_hd__nand2_1 _219_ (
234         .A(b[0]),
235         .B(a[0]),
236         .Y(_151_)
237     );
238     sky130_fd_sc_hd__xor2_1 _220_ (
239         .A(b[0]),
240         .B(a[0]),
241         .X(_152_)
242     );
243     sky130_fd_sc_hd__and3b_1 _221_ (
244         .A_N(sel[0]),
245         .B(sel[1]),
246         .C(sel[2]),
247         .X(_153_)
248     );
249     sky130_fd_sc_hd__nand3b_1 _222_ (
250         .A_N(sel[0]),
251         .B(sel[1]),
252         .C(sel[2]),
253         .Y(_154_)
254     );
255     sky130_fd_sc_hd__nor2_1 _223_ (
256         .A(sel[0]),
257         .B(sel[1]),
258         .Y(_155_)

```


```

1581     sky130_fd_sc_hd__a31oi_1 _443_ (
1582         .A1(b[7]),
1583         .A2(a[7]),
1584         .A3(_160_),
1585         .B1(_147_),
1586         .Y(_149_)
1587     );
1588     sky130_fd_sc_hd__a211oi_1 _444_ (
1589         .A1(_158_),
1590         .A2(_130_),
1591         .B1(_146_),
1592         .C1(_148_),
1593         .Y(_150_)
1594     );
1595     sky130_fd_sc_hd__a31oi_1 _445_ (
1596         .A1(sel[3]),
1597         .A2(_149_),
1598         .A3(_150_),
1599         .B1(_144_),
1600         .Y(y[7])
1601     );
1602 endmodule

```



LOGIC SYNTHESIS: EXAMPLE ALU YOSYS OUTPUT



LOGIC SYNTHESIS: EXAMPLE SEQUENTIAL

```


DFF.v
1 module DFF (CLK, RST, EN, D, Q);
2   input      CLK, RST, EN;
3   input      D;
4   output reg Q;
5
6   always @(posedge CLK or posedge RST)
7   begin
8     if (RST == 1'b1)
9       Q <= 0;
10    else if (EN == 1'b1)
11      Q <= D;
12  end
13
14 endmodule
15

```

```

DFF_TB.v
1 `timescale 1 ns / 1 ns
2
3 module main;
4   reg CLK, RST, EN;
5   reg D;
6   wire Q;
7
8   DFF dut(.CLK(CLK), .RST(RST), .EN(EN), .D(D), .Q(Q));
9
10  // Test clock generation.
11  always
12    #25 CLK = !CLK;
13
14  initial
15  begin
16    CLK = 1'b0;
17    #10 D = 1'b0; EN = 1'b0; RST = 1'b1;
18    #50 D = 1'b0; EN = 1'b1; RST = 1'b0;
19    #50 D = 1'b1;
20    #50 D = 1'b0;
21    #50 D = 1'b1; EN = 1'b0;
22    #50 D = 8'b0; EN = 1'b1;
23    #50 D = 8'b0;
24    #50 D = 8'b1;
25    #50 D = 1'b0;
26    $finish;
27  end
28
29  initial
30  $monitor("%t: RST=%b, EN=%b, D=%b, Q=%b", $time, RST, EN, D, Q);
31
32  initial
33  begin
34    $dumpfile("waveforms.vcd");
35    $dumpvars(0,dut);
36  end
37
38 endmodule

```



LOGIC SYNTHESIS: EXAMPLE SEQUENTIAL


```

DFF.v
DFF_map.v

1 /* Generated by Yosys 0.45+139 (gi
2
3 module DFF_map(CLK, RST, EN, D, Q);
4   wire _0_;
5   wire _1_;
6   input CLK;
7   wire CLK;
8   input D;
9   wire D;
10  input EN;
11  wire EN;
12  output Q;
13  wire Q;
14  input RST;
15  wire RST;
16  sky130_fd_sc_hd__clkinv_1 _2_ (
17    .A(RST),
18    .Y(_0_)
19  );
20  sky130_fd_sc_hd__mux2_1 _3_ (
21    .A0(Q),
22    .A1(D),
23    .S(EN),
24    .X(_1_)
25  );
26  sky130_fd_sc_hd__dfrtp_1 _4_ (
27    .CLK(CLK),
28    .D(_1_),
29    .Q(Q),
30    .RESET_B(_0_)
31  );
32 endmodule
33

```

**Yosys synthesized a DFF from the library:
However, the reset is logic_low for the
DFF in the library. So, it inserted an
inverter for the reset.
The enable is handled by adding a mux at
the input of the DFF**



DFF

